Pendant mon stage, j'ai travaillé sur SMTCoq [1, 9] qui sert aussi d'interface à différents prouveurs automatiques mais qui a la particularité de reproduire fidèlement, en Coq, le certificat reçu. Cette approche nécessite la mise en place d'un vérificateur des certificats fournis par les prouveurs automatiques et permet donc, par la même occasion, d'améliorer la confiance que l'on a dans ces outils.

Nous aimerions que le développement qui s'appuie sur cette automatisation soit adapté aux assistants de preuve dans lesquels les preuves sont modulaires et peuvent reposer sur des lemmes précédemment démontrés. Nous voudrions aussi pouvoir formaliser une théorie en partant des axiomes de celle-ci puis démontrer automatiquement de nouvelles propriétés de cette théorie.

1.3 Contribution proposée

Pendant mon stage, je me suis donc attaché à améliorer l'expressivité de SMTCoq en permettant le développement incrémental d'une théorie. En effet, j'ai ajouté la possibilité de transmettre au prouveur automatique, au sein de SMTCoq, des axiomes ou des lemmes déjà démontrés. Cet ajout s'est traduit par une extension de la logique utilisée et se retrouve dans toutes les étapes intermédiaires de SMTCoq, étapes que j'ai étendues en conséquence. Cette extension a nécessité un développement original: la technique d'encodage des instanciations des lemmes que j'ai utilisée (7.5) permet d'alléger la suite de la vérification. Enfin, j'ai automatisé le procédé de vérification final afin de préserver la facilité d'utilisation de SMTCoq.

1.4 Arguments en faveur de la validité de la contribution

En plus des tests présents dans SMTCoq initialement, le code final passe d'autres tests qui requièrent la transmission de lemmes au prouveur automatique. Cela atteste que l'implémentation de la méthode est valide et que celle-ci induit un gain d'expressivité. J'ai ainsi pu vérifier des propriétés, automatiquement et dans Coq, portant sur: la théorie des groupes, une théorie formalisant les listes d'entiers, des fonctions définies récursivement, etc.
D'autre part, ma contribution respecte le principe sceptique de SMTCoq (5.1): le développement et le calcul sont faits principalement en OCaml, en dehors de l'assistant de preuve. Cet aspect donne une meilleure robustesse à SMTCoq face aux changements internes des prouveurs automatiques.

1.5 Bilan et perspectives

Les formules acceptées par SMTCoq doivent être exprimées dans le type des booléens et doivent aussi être en forme prénexe. On pourrait améliorer l'expressivité en étendant les cas d'application (5.2.3) aux formules exprimées dans le type des propositions en Coq.
On pourrait aussi utiliser des méthodes de *machine learning* pour sélectionner les lemmes à envoyer au prouveur automatique comme c'est fait dans Coqhammer et Sledgehammer [4, 6]. L'avantage étant qu'avec des lemmes pertinents et en petit nombre le prouveur automatique trouve plus rapidement et plus souvent la preuve du théorème en question.
Un autre but que l'on souhaite poursuivre est la certification du logiciel de vérification Why3 [7]. Puisque Why3 utilise des prouveurs automatiques, cela nécessite de certifier les démonstrations faites par ces prouveurs. Ce sujet est au cœur de celui de ma thèse intitulée "Certification de la génération et la transformation d'obligations de preuves" qui sera encadrée par Claude Marché, Chantal Keller et Andrei Paskevich. À cette occasion, j'utiliserai SMTCoq et je profiterai de l'amélioration de son expressivité due à mon stage. Je pourrai l'améliorer encore en étendant le format des lemmes que l'on peut transmettre à SMTCoq.

Ces améliorations de l'expressivité et de l'efficacité combinées avec la facilité d'utilisation et la robustesse de SMTCoq ont pour objectifs d'en faire un outil accessible et de généraliser son utilisation dans les projets développés en Coq.

2 Logiciels utilisés

2.1 Assistants de preuve

Les assistants de preuve sont des outils puissants qui permettent d'exprimer des théorèmes complexes puis de les vérifier de manière interactive. Ils proposent à un utilisateur de formuler son problème puis de le démontrer, le rôle principal de l'assistant de preuve étant alors de vérifier que la preuve fournie est correcte. Pour une propriété donnée, l'utilisateur doit donc construire une preuve parfaitement rigoureuse et exhaustive de la propriété ce qui peut rendre le processus de vérification long et fastidieux. La confiance accordée à ces outils dépend de la compréhension que l'on peut avoir de son noyau, la vérification étant effectuée par celui-ci. Afin de faciliter cette compréhension, le noyau de l'assistant de preuve Coq est implémenté en OCaml, un langage de haut niveau proche de sa logique. L'accent est mis sur la concision et la clarté du code.

Dans la suite nous utiliserons Coq comme assistant de preuve. La logique de son noyau se fonde sur le calcul des constructions inductives [13]. La sémantique du langage n'est pas donnée en détail dans ce rapport mais les aspects importants seront précisés au moment de leur utilisation. La partie 3 introduit deux techniques spécifiques à Coq que nous utiliserons.

2.2 Prouveurs automatiques

Les prouveurs automatiques, quant à eux, ne demandent pas de preuves de la part de l'utilisateur. L'effort de certification est alors réduit à la formalisation du problème et dans certains cas le prouveur donne une trace de son exécution appelée certificat. En contrepartie, la logique d'un prouveur automatique est plus limitée et/ou la réponse en temps fini n'est pas garantie. Puisqu'un prouveur automatique doit chercher la preuve du théorème entré, l'efficacité de son implémentation est capitale. Pour cette raison, les prouveurs automatiques sont écrits dans des langages de plus bas niveau tels que C ou C++ et font usage de structures mutables complexes.

La partie 4 détaille le fonctionnement des prouveurs automatiques ainsi que l'utilisation que nous en ferons.

2.3 SMTCoq

Une interface entre assistant de preuve et prouveurs automatiques telle que SMTCoq offre les avantages des deux types de logiciels formels décrits ci-dessus. Un autre avantage de SMTCoq est sa modularité: il est en effet possible de prendre en charge d'autres prouveurs automatiques, ceux-ci n'étant pas nécessairement du même type (prouveurs SAT/SMT) et n'ayant pas nécessairement le même format d'entrée ni le même format de sortie.

La partie 5 présente SMTCoq qui est l'objet du stage et qui est un projet actuellement développé par Chantal Keller en collaboration avec l'Université de l'Iowa. Le logiciel SMTCoq est écrit en OCaml et en Coq, langages que j'ai donc utilisés pendant mon stage.

Les fragments de code Coq fournis sont encadrés et sont compilables lorsqu'ils sont chargés dans l'environnement de SMTCoq, c'est-à-dire lorsqu'ils sont précédés de:

```
Require Import SMTCoq Bool.
Open Scope Z_scope.
```

Cela requiert l'installation de SMTCoq, la version contenant ma contribution est disponible à l'adresse:

`https://github.com/QGarchery/smtcoq-1`

On préférera la version utilisant native-coq [2] pour des raisons d'efficacité.

3 Techniques de preuves en Coq

Cette partie présente deux techniques de preuves en Coq qui sont combinées dans SMTCoq: la réflexion calculatoire et la réification.

3.1 Réflexion calculatoire

Il est possible d'introduire de nouveaux termes Coq de manière logique à l'aide d'un type inductif ou de manière calculatoire avec une définition. Nous allons voir que la technique de réflexion calculatoire n'est possible que dans le second cas. Nous commençons par détailler les types inductifs, ce qui permettra de mettre en avant cette différence.

3.1.1 Types inductifs

Prenons l'exemple des formules disjonctives booléennes. Ces formules sont des éléments du type inductif `OrTree` qui a un constructeur `Bool` pour les feuilles et un constructeur `Or` pour les nœuds:

```
Inductive OrTree :=
  Bool (b : bool)
| Or (left: OrTree) (right: OrTree).
```

Une première manière de donner l'interprétation des formules est d'utiliser un type inductif:

```
Inductive Interp : OrTree -> Prop :=
  InterpBool :
    Interp (Bool true)
| InterpOrLeft t1 t2 :
    Interp t1 -> Interp (Or t1 t2)
| InterpOrRight t1 t2 :
    Interp t2 -> Interp (Or t1 t2).
```

Pour tout arbre `t`, `Interp t` signifie que l'interprétation de `t` est vraie. Le cas des feuilles `true` est traité par `InterpBool`. Le constructeur `InterpOrLeft` nous assure que si l'interprétation de `t1` est vraie alors, pour tout arbre `t2`, l'interprétation de `Or t1 t2` est vraie. Le cas `InterpOrRight` est similaire.

On peut alors faire des preuves sur les éléments de ce type inductif:

```
Definition t := Or (Or (Bool false) (Bool true)) (Or (Bool false) (Bool false)).

Lemma Interp_t : Interp t.
Proof.
  apply InterpOrLeft.
  apply InterpOrRight.
  apply InterpBool.
Qed.
```

3.1.2 Définitions en Coq et convertibilité

La réflexion calculatoire repose sur la convertibilité de deux termes: deux termes sont convertibles lorsqu'ils se réduisent vers un même terme. Aussi, à chaque nouvelle définition, la nouvelle constante qui est définie est convertible à sa définition.

Plutôt que d'utiliser un prédicat qui prend une formule disjonctive `t` et qui énonce que l'interprétation de `t` est vraie, on peut définir une fonction récursive qui calcule l'interprétation et combiner cette fonction avec le prédicat de l'égalité. On peut ensuite montrer que les deux formalisations sont bien équivalentes.

```
Fixpoint interp (t : OrTree) :=
  match t with
    Bool b => b
  | Or t1 t2 => interp t1 || interp t2
  end.

Proposition Interp_equiv_interp t :
  Interp t <-> interp t = true.
```

On a noté `||` la fonction Coq `orb` qui implémente la disjonction booléenne. L'avantage de cette définition de l'interprétation par rapport au type inductif est que, grâce à la convertibilité de `interp` à sa définition, on a une preuve triviale du lemme précédent.

```
Lemma interp_t : interp t = true.
Proof.
  reflexivity.
Qed.
```

En effet, en notant `a` ≡ `b` lorsque `a` est convertible à `b`, on a:

```
interp t ≡ interp (Or (Bool false) (Bool true)) ||
           interp (Or (Bool false) (Bool false))
         ≡ (interp (Bool false) || interp (Bool true)) ||
           (interp (Bool false) || interp (Bool false))
         ≡ (false || true) || (false || false)
         ≡ true
```

Ce fonctionnement peut être exploité pour construire des preuves qui reposent sur un calcul de convertibilité de termes Coq.

3.2 Réification

3.2.1 *Embeddings*

La réification est le fait de passer d'un *shallow-embedding* à un *deep-embedding*. Dans le cas du *deep-embedding*, un terme est représenté dans un type de données du langage ce qui met en évidence sa structure. À l'inverse, un *shallow-embedding* du même terme est traduit directement vers sa valeur dans le langage cible.

Reprenons l'exemple des formules disjonctives et considérons la formule $u := (b_1 \vee b_2) \vee (b_3 \vee b4)$. À l'instar de la section précédente, le *deep-embedding* de u est donné dans le type `OrTree`:

```
Or (Or (Bool b1) (Bool b2)) (Or (Bool b3) (Bool b4))
```

et son *shallow-embedding* peut être donné en utilisant la disjonction booléenne de Coq:

```
(b1 || b2) || (b3 || b4)
```

Dans la suite, on s'intéresse au problème de mettre des formules booléennes disjonctives en forme de peigne. Avec `b1`, `b2`, `b3` et `b4` des termes Coq de type `bool`, on veut, par exemple, pouvoir passer de

`v := (b1 || b2) || (b3 || b4)` à `v' := b1 || (b2 || (b3 || b4))`

Pour cela, on a besoin de récupérer la structure du booléen `v`, c'est l'étape de réification. Il s'agit donc de construire, à partir de `v`, un terme du type `OrTree` qui a la même structure que `v`.

3.2.2 Méthodes

Il n'est pas possible d'écrire dans le langage du noyau de Coq une fonction qui transforme un terme booléen en forme disjonctive en l'élément du type `OrTree` correspondant. En effet, le type `bool` n'ayant que deux constructeurs (`true` et `false`), inspecter par *pattern-matching* ne pourra donner que l'un de ces deux constructeurs.

Une solution est de travailler directement avec le type de données OCaml des termes Coq. C'est l'approche utilisée par SMTCoq.

Le méta-langage des tactiques de Coq, langage non typé, non terminant et implémenté en dehors du noyau, offre la possibilité de manipuler la représentation des termes du noyau. Dans le cas des formules disjonctives, on écrit:

```
Ltac reify A := match A with
  | orb ?X ?Y => let rx := reify X in
                 let ry := reify Y in
                 constr:(Or rx ry)
  | ?X => constr:(Bool X) end.
```

Le terme `u := (b1 || b2) || (b3 || b4)` réifié donne bien

`Or (Or (Bool b1) (Bool b2)) (Or (Bool b3) (Bool b4))`

et on notera que l'interprétation de ce nouveau terme est convertible à `u`.

3.2.3 Intérêt et exemple d'utilisation

L'intérêt de la réification est que la structure du terme réifié est mise en évidence. Il devient alors possible de manipuler explicitement cette structure.

Par exemple, sur le type `OrTree`, il est possible de définir une fonction `peigne` qui renvoie l'arbre en argument mis sous forme de peigne. Le théorème de correction de cette fonction établit que

`forall t : OrTree, interp (peigne t) = interp t`

Ce théorème de correction n'est applicable que si on a un terme de la forme `interp x`. Si `u` est un terme booléen dont la réification est `v` (de type `OrTree`), on peut mettre `u` sous cette forme en remarquant que `interp v` $\equiv$ `u`.

En combinant tous ces résultats, on peut définir une tactique `peignify` qui met en forme de peigne les formules disjonctives des deux côtés de l'égalité. Cette tactique permet par exemple de démontrer le lemme suivant:

```
Lemma or_tree_equality b1 b2 b3 b4:
  (b1 || b2) || (b3 || b4) = b1 || ((b2 || b3) || b4).
Proof.
  peignify. reflexivity.
Qed.
```

Cette preuve a un contenu calculatoire: le calcul de la fonction `peigne` sur la réification des termes booléens de chaque côté de l'égalité. Pour le code complet de cette partie et en particulier le code de la tactique `peignify`, voir l'annexe A, la preuve de correction est inspirée de [13], section 3.3.

4 Prouveurs automatiques

4.1 Utilisation par SMTCoq

Différents prouveurs automatiques sont mis à la disposition de l'utilisateur de SMTCoq. Parmi ceux-ci, il y a zChaff, un prouveur SAT, c'est-à-dire un prouveur qui résout des problèmes de satisfiabilité de formules booléennes. On trouve aussi des prouveurs SMT (veriT et CVC4) que nous présentons plus en détail dans cette partie.

Pour SMTCoq, les prouveurs automatiques sont vus comme des boîtes noires qui résolvent des problèmes logiques. Plus précisément, SMTCoq interagit avec un prouveur automatique en traduisant le problème dans un format reconnu par celui-ci (4.3) puis en interprétant sa réponse (4.4). Ce fonctionnement est un des intérêts de l'approche sceptique (5.1).

4.2 Prouveurs SMT

Un problème SMT est un problème de satisfiabilité de formules du premier ordre pour certaines théories prédéfinies. L'algorithme DPLL [12] de résolution des problèmes SAT peut être combiné avec une procédure de décision d'une théorie pour donner un prouveur SMT. Expliquons succinctement le fonctionnement d'une version de cet algorithme DPLL. Nous ne considérerons que la théorie LIA, théorie qui contient les entiers et les symboles d'addition, de soustraction et d'inégalité. On prend en exemple le problème qu'on appellera *pb_lia* et qui consiste à satisfaire la formule

$$((x + y \leq -3 \wedge y \geq 0) \vee x \leq -3) \wedge x \geq 0$$

où x et y sont des entiers.

L'algorithme commence par identifier les atomes, c'est-à-dire les sous-formules spécifiques à la théorie LIA. Dans notre exemple cela revient à poser

$$a := \ x + y \leq -3 \qquad c := \ x \leq -3$$
$$b := \ y \geq 0 \qquad d := \ x \geq 0$$

et à chercher si la formule $((a \wedge b) \vee c) \wedge d$ est satisfiable. Cette étape peut être suivie par une transformation de Tseitin (6.2) afin d'obtenir un problème en Forme Normale Conjonctive (CNF) qui est le format de prédilection des prouveurs SAT.

Ensuite, l'algorithme DPLL répète les étapes suivantes:

- Appeler un prouveur SAT sur la formule. Si ce n'est pas satisfiable, alors l'algorithme répond que le problème n'est pas satisfiable.
- Dans le cas contraire, le prouveur SAT donne une instanciation qui satisfait toutes les formules. La formule de l'exemple est satisfiable, une instanciation pourrait être: $a \wedge \neg b \wedge c \wedge d$, ce qui signifie que toutes les variables booléennes sont à $true$ sauf b qui est à $false$.
- Appeler la procédure de décision pour vérifier que l'instanciation est valide dans la théorie. Si c'est le cas, l'algorithme renvoie que le problème est satisfiable.
- Sinon, rajouter la négation de l'instanciation à la liste des formules et recommencer. C'est le cas de notre exemple: il n'est pas possible, dans la théorie LIA, d'avoir à la fois $x \leq -3$ et $x \geq 0$. À la fin de cette étape, le problème est donc ramené à la satisfiabilité des deux formules suivantes:

$$((a \wedge b) \vee c) \wedge d$$
$$\neg a \vee b \vee \neg c \vee \neg d.$$

Cet algorithme termine puisqu'à chaque répétition des étapes ci-dessus une nouvelle instanciation des atomes n'est plus possible et qu'il y a un nombre fini de telles instanciations.

4.3 Format d'entrée, le langage SMT-LIB

Le langage SMT-LIB a vocation à être un format d'entrée commun à différents prouveurs SMT tels que veriT. Ce langage offre donc un cadre pour faire des comparaisons de prouveurs SMT. La définition du langage [3] donne des indications sur la sémantique que doivent avoir certaines constructions du langage.

Le problème *pb_lia* de la section précédente peut être écrit dans le langage SMT-LIB:

```
(set-logic QF_LIA)
(declare-fun x () Int)
(declare-fun y () Int)
(assert (and (or (and (<= (+ x y) (- 3)) (>= y 0)) (<= x (- 3))) (>= x 0)))
(check-sat)
(exit)
```

On appelle assertion une formule contenue dans une ligne d'un fichier SMT-LIB commençant par *assert*: la formule $((x + y \leq -3 \wedge y \geq 0) \vee x \leq -3) \wedge x \geq 0$ est une assertion dans le fichier ci-dessus.

4.4 Format de sortie, les certificats de veriT

Lorsqu'on appelle un prouveur automatique sur un fichier SMT-LIB, si la conjonction des assertions est satisfiable, le prouveur renvoie *sat*. Si la conjonction des assertions n'est pas satisfiable, le prouveur renvoie *unsat* et fournit un fichier de certificat expliquant la raison de cette non-satisfiabilité. Le format de ce fichier de certificat peut varier en fonction du prouveur SMT considéré.

Dans la suite de cette section nous nous intéressons aux certificats du prouveur SMT veriT. Ceux-ci utilisent des clauses, une clause étant une disjonction de formules écrite sous forme de liste. Nous noterons (f_1 ... f_n) la clause contenant les formules f_1, ..., f_n. La clause vide, notée (), représente l'absurde. Les certificats de veriT sont constitués d'une suite de règles de la forme:

$$id : (typ\ cl\ dep)$$

où *id* est un entier qui identifie la règle, *typ* est le type de la règle, *cl* est une clause qu'on appelle résultat de la règle et *dep* liste toutes les dépendances de la règle. Ainsi, montrer que le problème n'est pas satisfiable revient à obtenir une suite de règles dont le résultat de l'une d'entre elles est la clause vide. Une règle peut utiliser le résultat d'une autre règle identifiée par *id*, dans ce cas sa liste de dépendances contient l'identifiant *id*.

4.4.1 Règle *input*

Les règles de type *input* sont des règles qui établissent les hypothèses du problème et correspondent aux assertions du fichier SMT-LIB. Par exemple, la règle

$$1 : (input\ (x \geq 0))$$

est utilisée dans le cas où on suppose que x est un entier positif. Cette règle (et les règles *input* en général) ne dépend pas du résultat d'autres règles.

4.4.2 Règle *resolution*

Pour manipuler ces clauses, veriT peut utiliser une règle *resolution*. La règle de résolution appliquée à deux clauses donne une nouvelle clause contenant toutes les formules de ces deux clauses sauf les formules qui ont leur négation dans l'autre clause. Par exemple, la résolution de $a \vee b$ et de $\neg a \vee \neg c$

donne $b \vee \neg c$. Grâce à cette règle, on peut montrer que le problème qui suppose l'existence d'un entier x tel que $x \geq 0$ et $\neg(x \geq 0)$ n'est pas satisfiable:

$$1 : (input\ (x \geq 0))$$
$$2 : (input\ (\neg(x \geq 0)))$$
$$3 : (resolution\ ()\ 1\ 2)$$

Une règle de résolution peut avoir plus de deux dépendances. Son résultat est alors défini récursivement à partir de ceux des règles de résolution appliquées à deux clauses. Par exemple, le résultat de la règle

$$4 : (resolution\ (B)\ 1\ 2\ 3)$$

est le même que celui de la règle 5 lorsque la règle ci-dessus est remplacée par les règles

$$4 : (resolution\ (A)\ 1\ 2)$$
$$5 : (resolution\ (B)\ 4\ 3)$$

4.4.3 Règles *not_implies*0 et *not_implies*1

La règle *not_implies*0 implémente la règle logique qui donne A à partir de $\neg(A \Rightarrow B)$ et *not_implies*1 implémente la règle logique qui donne $\neg B$ à partir de $\neg(A \Rightarrow B)$.

Ces règles dépendent donc du résultat d'une autre règle qui doit nécessairement être une clause de la forme $(\neg(X \Rightarrow Y))$. Les résultats des règles *not_implies*0 et *not_implies*1 sont alors respectivement (X) et $(\neg Y)$. Cette règle nous permet de montrer que la formule $\neg(A \Rightarrow A)$ n'est pas satisfiable:

$$1 : (input\ (\neg(A \Rightarrow A)))$$
$$2 : (not\_implies0\ (A)\ 1)$$
$$3 : (not\_implies1\ (\neg A)\ 1)$$
$$4 : (resolution\ ()\ 2\ 3)$$

5 Présentation de SMTCoq

5.1 SMTCoq, une interface sceptique entre Coq et les prouveurs automatiques

Pour améliorer l'automatisation de Coq et y intégrer l'utilisation de prouveurs automatiques, il y a principalement deux approches.



L'approche autarcique consiste à vérifier le code du prouveur automatique à l'intérieur de l'assistant de preuve. L'avantage de cette méthode est qu'une fois cette vérification faite, on sait que chaque appel du prouveur automatique nous renverra une preuve correcte.

Dans l'approche sceptique, le certificat renvoyé par le prouveur automatique est vérifié à chaque appel de celui-ci. Cette approche, utilisée par SMTCoq, ne permet pas de garantir la complétude du système: certains buts valides ne sont pas démontrés, notamment lorsque le prouveur automatique renvoie un certificat erroné ou que la reconstruction de la preuve par SMTCoq n'est pas possible. En revanche, cette approche ne fige pas l'implémentation du prouveur automatique puisque ce n'est pas son code qui est vérifié mais sa réponse. Un autre avantage est que l'effort de certification est plus restreint: pour un certificat fixé, il faut vérifier que celui-ci correspond bien à une preuve du but.

5.2 Fonctionnement de SMTCoq

5.2.1 Amélioration de l'automatisation

Chacune des tactiques Coq définies par SMTCoq invoque un prouveur automatique différent: zChaff, CVC4 ou veriT. Ces tactiques permettent à l'utilisateur Coq de faire appel à un prouveur automatique pour résoudre le but courant et donc de profiter de l'automatisation du prouveur.

Les prouveurs automatiques fournissent un certificat uniquement dans le cas où le problème n'est pas satisfiable (4.4). Pour exploiter ce fonctionnement, SMTCoq envoie la négation du but au prouveur automatique. La preuve ne peut être reconstruite que dans le cas où la réponse est *unsat* et est accompagnée d'un fichier de certificat. Ce fichier prouve que la négation du but mène à l'absurde. Autrement dit, on obtient une preuve de la double négation du but. Afin de pouvoir en déduire le but, SMTCoq requiert des formules exprimées dans les booléens. En effet, le type des booléens relève de la logique classique alors que le type des propositions de Coq est intuitionniste.



La première étape est la réification, l'énoncé du lemme est traduit en un AST appartenant au type des formules acceptées par SMTCoq. Cet AST est traduit en un fichier SMT-LIB qui est donné en entrée au prouveur automatique. En cas de succès de ce prouveur, on obtient un fichier de certificat. Ce fichier de certificat est ensuite traduit par le préprocesseur dans un format adapté au vérificateur de SMTCoq. Ce dernier rejoue la preuve en Coq (voir partie 6).

5.2.2 Amélioration de la confiance

Dans la suite on s'attachera principalement à développer l'aspect automatisation de Coq mais SMTCoq propose également une commande de reconstruction d'une preuve effectuée par un prouveur automatique.



Cette commande prend en paramètre un fichier *fsmt* décrivant le lemme (typiquement écrit en SMT-LIB) et le certificat fourni par un prouveur automatique. Le lemme Coq est reconstruit à l'aide de l'interpréteur de SMTCoq et la preuve est reconstruite grâce au vérificateur. Une fois la reconstruction faite, la vérification que la preuve correspond bien au lemme est laissée à Coq.

Puisqu'un nouveau lemme Coq est créé, l'utilisateur peut vérifier que c'est bien le but qu'il voulait prouver. Ainsi, la confiance dans les prouveurs automatiques est améliorée: on peut vérifier la réponse du prouveur.

5.2.3 Cas d'application de SMTCoq

Dans les deux cas, les formules acceptées sont les formules logiques booléennes en forme prénexe. Les formules atomiques doivent aussi être décidables. Les théories suivantes et leurs combinaisons sont implémentées dans SMTCoq: arithmétique linéaire sur $\mathbb{Z}$, égalité et fonctions non interprétées, auxquelles s'ajouteront la théorie des vecteurs de bits et la théorie des tableaux.

5.3 Utilisation de SMTCoq

5.3.1 La tactique `verit`

Afin d'améliorer l'automatisation de Coq (5.2.1), SMTCoq définit la tactique `verit`. Cette nouvelle tactique permet de résoudre automatiquement les buts dans les booléens en forme prénexe. On reprend l'exemple de la partie 4. En utilisant les fonctions booléennes de Coq, `negb` pour la négation, `&&` pour la conjonction, et `>=?` et `<=?` pour les inégalités larges, le problème *pb_lia* devient :

```
Lemma pb_lia :
  forall x y,
    negb ( ((x+y <=? - (3)) && (y >=? 0) || (x <=? - (3))) && (x >=? 0)).
Proof.
    verit.
Qed.
```

La tactique `verit` commence par introduire les variables quantifiées universellement en tête de formule (dans l'exemple ce sont `x` et `y`) puis s'attend à ne pas avoir d'autres quantificateurs. C'est ensuite la négation de la formule qui est envoyée à veriT. La reconstruction de la preuve ne peut avoir lieu que si veriT renvoie *unsat* ainsi qu'un fichier de certificat.

5.3.2 La commande de reconstruction

La commande `Verit_Theorem` nous permet de créer un terme Coq à partir du certificat fourni par veriT appelé sur le problème *pb_lia*. On obtient alors le terme Coq `pb_lia`:

```
U:---  *goals*        All (1,0)      (Coq Goals Wrap)
1 pb_lia
2      : negb
3          (((Smt_var_x + Smt_var_y <=? - (3)) && (Smt_var_y >=? 0)
4            || (Smt_var_x <=? - (3))) && (Smt_var_x >=? 0))
```

On notera que ce terme représente bien la négation de la formule énoncée dans le fichier SMT-LIB, ce qui nous permet de vérifier la réponse du prouveur automatique (5.2.2).

6 Transformation de certificats

Nous avons vu que c'est grâce à un procédé de réification (3.2) que l'on peut transmettre le but Coq initial au prouveur automatique. Nous allons maintenant voir comment la réponse du prouveur automatique, le fichier de certificat, peut être transformée en un terme de preuve Coq.

6.1 Des certificats de toutes les couleurs

La tactique `verit` interprète le fichier de certificat *tcertif* fourni par le prouveur automatique en un terme Coq *ccertif*, ce qui permet de construire une preuve du but initial.

Format	Fichier texte	Code OCaml	Code Coq
Appellation	*tcertif*	*ocertif*	*ccertif*
Composant	*trule*	*orule*	*crule*

Cette interprétation passe par une étape intermédiaire *ocertif* écrite en OCaml. Cette étape a plusieurs avantages. En premier lieu, elle permet d'utiliser les outils de *parsing* du fichier de certificat (*ocamllex*, *ocamlyacc*). Par ailleurs, en utilisant la représentation OCaml des termes Coq, la traduction d'un *ocertif* en un *ccertif* est facilitée. Enfin, les *ocertif* sont définis dans un format facilement manipulable ce qui permet d'appliquer des adaptations (7.3.1 et 7.5), des simplifications (6.3.1) ou encore des optimisations (6.4.2).

6.2 Transformation de Tseitin et *hash consing*

6.2.1 Motivations

La transformation de Tseitin d'une formule donne une formule équisatisfiable qui est en CNF en introduisant de nouvelles variables booléennes. L'avantage de cette transformation est que sa complexité est linéaire en temps comme en espace. En comparaison, l'utilisation des lois de De Morgan pour obtenir une formule en CNF a une complexité en pire cas exponentielle. Le principe de cette transformation est d'introduire de nouvelles variables pour toutes les sous-formules, ce qui correspond au *hash consing* qui est fait par SMTCoq [1]. Cette étape intervient au moment du *parsing*, c'est-à-dire au moment du passage d'un *tcertif* à un *ocertif*. Du point de vue de SMTCoq, ce procédé signifie un gain en espace (les sous-formules ne sont pas répétées) et un gain en temps (les comparaisons de formules deviennent des comparaisons d'entiers).

6.2.2 Fonctionnement

La transformation commence par nommer toutes les sous-formules en partant des feuilles. La nouvelle formule à satisfaire est alors la conjonction de la variable représentant toute la formule et de formules additionnelles qui garantissent que les nouvelles variables sont équivalentes aux sous-formules qu'elles représentent.

On reprend l'exemple de la partie 4. En nommant les atomes on avait obtenu la formule $((a \wedge b) \vee c) \wedge d$. On introduit également e pour la sous-formule $a \wedge b$, f pour la sous-formule $e \vee c$ et g pour la sous-formule $f \wedge d$ qui est en fait toute la formule. La formule transformée devient donc $g \wedge D_e \wedge D_f \wedge D_g$ où les D_α sont des formules qui nous assurent que la variable α est équivalente à la formule qu'elle représente. On a par exemple $D_e = (\neg a \vee \neg b \vee e) \wedge (\neg e \vee a) \wedge (\neg e \wedge b)$ qui nous assure que la variable e est équivalente à $a \wedge b$.

Dans SMTCoq, au lieu de rajouter les formules D_α, les sous-formules sont enregistrées dans un tableau: la variable e est un indice du tableau et le tableau contient, à cet indice, la formule représentée par e ($a \wedge b$ dans notre exemple).

6.3 Le vérificateur

Le vérificateur de SMTCoq contient un type inductif *crule* qui représente les *trule* ainsi qu'une fonction `checker`. Nous allons définir ces termes Coq, voir comment ils implémentent la sémantique des certificats de veriT (4.4) et comment ils sont utilisés pour produire le terme de preuve.

6.3.1 Le type inductif *crule*

Chaque constructeur du type inductif *crule* représente une famille de *trule*. Le constructeur `Res` ne représente que la règle *resolution* mais d'autres constructeurs peuvent représenter différentes règles en fonction de leurs paramètres. Par exemple, le constructeur `ImmBuildProj` regroupe les règles *not_implies*0 et *not_implies*1 (4.4.3) et contient aussi un paramètre entier qui vaut `0` ou `1` et qui indique quelle est la *trule* représentée. Ce regroupement de règles au fonctionnement similaire permet, dans la suite, de simplifier le traitement de ces règles.

On appellera *ccertif* le type des certificats en Coq représentés par des listes de *crule*.

6.3.2 La fonction récursive `checker`

Pour enregistrer le résultat des règles précédentes on utilise un tableau de clauses appelé état, le premier emplacement ayant pour indice `1`. On rappelle qu'une clause représente la disjonction d'une liste de formules et est notée entre parenthèses. La fonction `checker` implémente l'application des règles du certificat et modifie donc l'état à chaque nouvelle *crule* rencontrée. Plus précisément, `checker` est une fonction Coq qui est définie récursivement sur son paramètre de type *ccertif* et qui a aussi un paramètre état. Le résultat des *crule* suit le fonctionnement des *trule*. Par exemple le résultat de `ImmBuildProj 0 1` est (X) si le premier emplacement du tableau d'état est de la forme $(\neg(X \Rightarrow Y))$. À chaque appel de `checker`, une nouvelle *crule* est consommée, son résultat est enregistré dans l'état.

Il n'y a pas de *crule* correspondant à la règle *input*: l'état est initialisé avec le résultat de la règle *input*. Dans un premier temps nous considérerons que la taille du tableau est égale au nombre de règles du *tcertif*.

Lorsque toute la liste *ccertif* a été consommée, `checker` renvoie `true` si le dernier emplacement modifié de l'état contient la clause vide et `false` sinon.

6.3.3 Théorème de correction

Sur le type des clauses, SMTCoq définit une fonction `interp` qui inverse la réification (3.2.3). Le vérificateur repose sur le théorème de correction qui nous assure que pour tout *ccertif* `cc` et toute clause `cl`, si le tableau d'état `t` est initialisé avec `cl`, alors:

```
checker cc t = true -> negb (interp cl)
```

Une fois celui-ci démontré, la preuve du but initial est obtenue en appliquant le théorème de correction puis la réflexivité de l'égalité. En effet, si `checker cc t` est convertible à `true`, alors l'hypothèse `checker cc t = true` est convertible à `true = true`.

6.3.4 Preuve du théorème de correction

On dit qu'une clause est valide lorsque son interprétation est vraie. La preuve du théorème de correction utilise la proposition `step_checker_correct` qui s'énonce ainsi: si toutes les clauses du tableau d'état sont valides alors une étape de `checker` modifie ce tableau en un tableau dont toutes les clauses sont valides. Il suffit en fait de vérifier que la nouvelle clause générée par `checker` est valide.

À partir de cette proposition on démontre le théorème de correction. Donnons les grandes étapes de la preuve de ce théorème. Pour prouver `checker cc t = true -> negb (interp cl)`, on suppose `interp cl` et `checker cc t = true`. Initialement, l'état ne contient donc que des clauses valides (tous les emplacements du tableau contiennent `cl`). D'après `step_checker_correct`, cette propriété est conservée par la fonction `checker`. Si de plus `checker` renvoie `true` cela veut dire qu'il y a la clause vide dans le dernier tableau d'état. L'interprétation de la clause vide nous conduit à une contradiction.

6.3.5 Exemple de l'identité

Considérons la proposition Coq suivante où `implb` est l'implication booléenne:

```
Proposition identity A : implb A A.
```

Lorsqu'on applique `verit`, puisque c'est la négation du but qui est envoyée, le prouveur automatique nous renvoie le *tcertif* de 4.4.3. Celui-ci est traduit par SMTCoq en un *ccertif*

```
cc := [ImmBuildProj 0 1; ImmBuildProj 1 1; Res 2 3]
```

et l'état est initialisé en un tableau

```
t := [|nid; nid; nid; nid|]
```

où `nid` est la clause $(\neg(A \Rightarrow A))$. Le calcul de `checker` est alors:

$$\begin{aligned}
\texttt{checker cc t} &\equiv \texttt{checker [ImmBuildProj 1 1; Res 2 3] [|nid;} \ (A)\texttt{; nid; nid|]} \\
&\equiv \texttt{checker [Res 2 3] [|nid;} \ (A)\texttt{;} \ (\neg A)\texttt{; nid|]} \\
&\equiv \texttt{checker [] [|nid;} \ (A)\texttt{;} \ (\neg A)\texttt{;} \ ()\texttt{|]} \\
&\equiv \texttt{true}
\end{aligned}$$

En appliquant le théorème de correction, on obtient `negb (interp` $(\neg(A \Rightarrow A))$`)` qui est convertible à `negb (negb (implb A A))`. Cette proposition est bien équivalente à `implb A A`.

6.4 Le préprocesseur

La transformation des certificats passe par une étape intermédiaire (6.1) qui nous permet d'améliorer l'efficacité de la transformation et facilite la traduction des certificats dans un format adapté au vérificateur. Nous donnons le format des certificats de cette étape puis une optimisation rendue possible par cette étape.

6.4.1 Le type OCaml *orule*

Les *orule* sont des enregistrements constitués, en plus du code OCaml d'une *crule*, de méta-données qui permettent l'implémentation d'un préprocesseur efficace.

6.4.2 Allocation dans le tableau d'état

On remarque que dans l'exemple de l'identité (6.3.5), le tableau d'état n'a pas besoin d'être de taille 4: seulement 2 emplacements sont nécessaires pour ce certificat. De manière générale, on peut réduire l'espace mémoire utilisé en calculant le nombre maximal de clauses à retenir à chaque étape de `checker`, c'est ce que fait la fonction *alloc*. Cette fonction assigne aussi, pour chaque règle, un emplacement dans l'état où enregistrer son résultat. Ce paramètre de position est noté au début des *crule*, de sorte que lorsque `checker` rencontre `ImmBuildProj 1 0 2`, le premier emplacement de l'état est modifié en (X) si le deuxième emplacement de l'état contient $(\neg(X \Rightarrow Y))$. Puisque les emplacements des clauses sont modifiés, il faut aussi modifier les dépendances des règles. Enfin, `checker` prend un paramètre qui indique quelle est la position du résultat de la dernière règle, celui-ci n'étant pas nécessairement au dernier emplacement de l'état comme précédemment. En initialisant `t` à `[|nid; nid|]` et en posant `cc := [ImmBuildProj 1 0 2; ImmBuildProj 2 1 2; Res 1 1 2]`, on a :

$$\begin{aligned}
\texttt{checker cc t 1} &\equiv \texttt{checker [ImmBuildProj 2 1 2; Res 1 1 2] [|}(A)\texttt{; nid|] 1} \\
&\equiv \texttt{checker [Res 1 1 2] [|}(A)\texttt{; }(\neg A)\texttt{|] 1} \\
&\equiv \texttt{checker [] [|}()\texttt{; }(\neg A)\texttt{|] 1} \\
&\equiv \texttt{true}
\end{aligned}$$

7 Préprocesseur pour les lemmes quantifiés

Dans cette partie on commence par donner la forme générale des certificats de veriT dans le cas des lemmes quantifiés universellement. On explique ensuite comment un certificat peut être modifié pour faciliter son encodage en un *ccertif* (partie 8).

Nous étudierons l'exemple suivant où `=?` est l'égalité à valeurs booléennes sur les entiers en Coq:

```
Lemma instance_2 f :
  (forall x, f (x+1) =? f x + 7) ->
  f 3 =? f 2 + 7.
```

7.1 Instanciation d'un lemme par veriT: la règle *forall_inst*

Lorsqu'un lemme en forme prénexe est donné par une règle *input*, veriT peut instancier ce lemme avec la règle *forall_inst*. L'appel de veriT sur l'exemple précédent fournit le certificat simplifié suivant:

$$\begin{aligned}
&1 : (input\ (\neg(f\ (3) = f\ 2 + 7))) \\
&2 : (input\ (\forall\ x,\ f\ (x+1) = f\ x + 7)) \\
&3 : (forall_inst\ (\neg(\forall\ x,\ f\ (x+1) = f\ x + 7)\ (f\ (2+1) = f\ 2 + 7))) \\
&4 : (resolution\ (f\ (2+1) = f\ 2 + 7)\ 3\ 2)
\end{aligned}$$

On remarquera que la règle d'instanciation *forall_inst* ne dépend d'aucune autre règle et que l'utilisation de son résultat se fait au moyen d'une règle *resolution*.

7.2 *Parsing* des certificats

Le *hash consing* des formules apparaissant dans le certificat (6.2) est à l'origine d'un gain en espace dans l'enregistrement de celles-ci. Ces formules sont ensuite interprétées pour construire le terme de preuve, il faut donc qu'elles soient traduites fidèlement. D'autre part, les variables qui apparaissent dans les lemmes et qui sont liées par leurs quantificateurs sont à traiter séparément des autres variables. En

effet, elles sont quantifiées universellement alors que les autres variables sont implicitement quantifiées existentiellement (c'est un problème de satisfiabilité). Par ailleurs ces variables n'ont pas de sens en dehors du lemme dans lequel elles sont quantifiées, il serait donc inutile d'enregistrer un terme contenant une de ces variables. Ce problème est résolu en maintenant, au moment du *parsing*, en plus de la formule qui est en train d'être traitée, une valeur booléenne qui indique si cette formule contient une variable quantifiée universellement. Lorsque ce booléen est à *true*, on n'enregistre pas la sous-formule. Par exemple dans la formule $\forall x, (f\ (x+1) = f\ x \wedge f\ 0 = 0)$, la sous-formule $f\ 0 = 0$ est enregistrée, mais ni $f\ (x+1) = f\ x$ ni $f\ (x+1) = f\ x \wedge f\ 0 = 0$ ne le sont.

7.3 Lier une instanciation à un lemme

7.3.1 Lien entre une règle *forall_inst* et une règle *input*

Dans l'exemple, la dépendance de la règle $forall\_inst$ au lemme donné dans la règle *input* est évidente pour plusieurs raisons: le certificat est de petite taille, il y a égalité syntaxique entre le lemme et une sous-formule du résultat de la règle $forall\_inst$, la règle de résolution qui utilise la règle $forall\_inst$ est située juste après celle-ci et a exactement 2 dépendances. Cependant, dans le cas général, aucune de ces raisons ne reste valide. En particulier, veriT fait un renommage des variables liées qui apparaissent dans les lemmes (le certificat complet est donnée dans l'annexe B). Retrouver la dépendance demande donc, a priori, d'unifier à α-équivalence près des formules contenues dans une règle $forall\_inst$ et dans les règles *input*. Heureusement, veriT fait un *hash consing* des formules qui apparaissent dans les *tcertif*. Cela nous permet d'enregistrer la dépendance au lemme dans la règle $forall\_inst$ au moment du *parsing* des certificats de veriT. La règle 3 devient:

$$3 : (forall\_inst\ (\neg(\forall\ x,\ f\ (x+1) = f\ x + 7)\ (f\ (2+1) = f\ 2 + 7))\ 2)$$

7.3.2 Lien entre une formule du certificat et un lemme Coq

On cherche à établir un lien entre une formule du certificat de veriT et un lemme rajouté par l'utilisateur de SMTCoq. La difficulté vient du fait que les lemmes additionnels peuvent apparaître modifiés dans les certificats de veriT. Rétablir la forme initiale est une solution trop coûteuse car il faudrait modifier la structure de toutes les formules suivantes qui en dépendent, directement ou indirectement. Pour résoudre ce problème, on utilise une nouvelle table de hachage.

Cette table, initialisée avec toutes les sous-formules des lemmes Coq, nous sert pour reconnaître des formules aux modifications de veriT près. Par exemple, une égalité peut apparaître inversée dans la règle *input* d'un lemme additionnel. À chaque fois qu'il faut enregistrer une nouvelle sous-formule de la forme $a = b$, en plus de vérifier si cette formule est déjà contenue dans la table, on regarde aussi si la formule $b = a$ est dans cette même table. Ainsi, on peut reconnaître efficacement des formules identiques modulo symétrie de l'égalité et retrouver à quel lemme se rapporte une formule.

7.4 Difficultés liées à la traduction de la règle *forall_inst*

On a vu que veriT utilise des règles d'intanciation des lemmes dont les résultats sont des clauses de la forme:

$$(\neg(\forall\, x,\ P\ x)\ (P\ n))$$

La traduction directe de ces règles pose plusieurs problèmes listés ci-dessous.

1. La clause ci-dessus est une tautologie pour tout prédicat P et toute valeur n en logique classique. Cependant, dans la logique intuitionniste de Coq, ce n'est plus vrai.

2. Initialement, seulement la règle *input* correspondant au but était prise en compte en initialisant l'état avec son résultat (6.3.2) mais on a vu que l'utilisation d'une règle d'instanciation dépend indirectement d'une règle *input* (7.1). Il faut donc créer une nouvelle *crule* pour représenter les règles *input* correspondant aux lemmes rajoutés par l'utilisateur.

3. Il faut modifier le format des *crule* pour accepter les formules quantifiées, ce qui demande ensuite de raisonner dans Coq sur des termes à α-équivalence près.

Une solution au problème (1) serait de remplacer les résultats des règles d'instanciation par des clauses de la forme

$$((\forall\, x,\ P\ x) \Rightarrow P\ n)$$

mais cela ne résout ni le problème (2) ni le problème (3) et demande donc de profonds changements des *ccertif* et de leur utilisation par le vérificateur.

7.5 Modifier le résultat de la règle $forall\_inst$

Pour ces raisons, il est préférable de modifier les règles de la forme:

$$id : (forall\_inst\ (\neg lemma\ lemma\_inst)\ id\_lemma)$$

où *lemma* est un des lemmes rajoutés par l'utilisateur et *lemma_inst* est une instance de ce même lemme en une règle:

$$id : (forall\_inst\ (lemma\_inst)\ id\_lemma)$$

Cette modification résout le problème (1) et le problème (3), la forme proposée ci-dessus ne faisant apparaître ni la forme logique de l'implication ni le lemme quantifié. Il faut aussi modifier les règles suivantes qui dépendent du résultat de cette nouvelle règle. On fait l'hypothèse supplémentaire qu'une règle $forall\_inst$ dépendant d'un lemme l ne sera utilisée dans la suite du certificat que dans une règle de résolution ayant aussi une dépendance à l. On se ramène donc à modifier seulement les règles de résolution, ce que l'on fait ainsi: si une règle de résolution a pour liste de dépendances *dep*, on trouve toutes les règles $forall\_inst$ de cette liste et on enlève leurs dépendances de *dep*. Cette modification a une complexité en pire cas linéaire dans la taille des certificats. Dans le cas où il ne reste plus qu'une seule dépendance, la règle $resolution$ devient une règle *same*, c'est-à-dire une règle qui a le même résultat qu'une règle précédente.

Par exemple, dans le certificat de l'exemple de cette partie (7.1), la règle 3 a pour dépendance la règle 2. La liste de dépendances de la règle de résolution contient l'identifiant 3 qui est une règle $forall\_inst$, il faut donc enlever la règle 2 de cette liste. La nouvelle liste de dépendances de la règle de résolution étant réduite à un seul élément, cette règle devient une règle *same*. Les règles 3 et 4 du certificat deviennent donc:

$$3 : (forall\_inst\ (f\ (2+1) = f\ 2+7)\ 2)$$
$$4 : (same\ (f\ (2+1) = f\ 2+7)\ 3)$$

7.6 Une application: les règles *input*

Cette modification est suffisamment générale pour permettre le traitement des règles *input*. On est effectivement amené à traiter ces règles lorsque l'utilisateur transmet des lemmes non quantifiés à la tactique `verit`. Le résultat d'une règle *input* d'un lemme non quantifié peut être directement utilisé par une des règles suivantes, ces lemmes n'ayant pas à être instanciés. Ainsi, les règles *input* qui ne sont ni des lemmes quantifiés ni la négation du but initial sont transformées en des règles *forall_inst* sans nécessiter de modifications des règles suivantes. Cette application résout partiellement le problème (2). La résolution complète de ce problème repose sur le lien établi entre une règle d'instanciation et le lemme Coq (7.3) et est présentée dans la partie suivante.

8 Vérificateur pour les lemmes quantifiés

Pour traiter le cas des lemmes quantifiés du point de vue de Coq, on a besoin de rajouter un constructeur au type inductif *crule*. On verra comment modifier le vérificateur en conséquence afin de rétablir la preuve de correction et de préserver l'automatisation de SMTCoq.

8.1 La *crule* `Forallinst`

Dans cette section, on se propose d'encoder les règles d'instanciation en ne rajoutant pas de quantificateurs dans les *crule*. Cet encodage n'est possible que pour les lemmes en forme prénexe; dans le cas général il faudrait nécessairement étendre le langage des termes, ce qui nous obligerait à étendre la fonction d'interprétation et à adapter les preuves la concernant. La légèreté de cet encodage est rendue possible par les modifications de la règle d'instanciation de la partie précédente. On rajoute tout de même un constructeur au type *crule* mais cela nous demande seulement de compléter la preuve de `step_checker_correct` (6.3.4) correspondant à cette nouvelle règle.

Le nouveau constructeur s'écrit:

```
Forallinst p lemma plemma lemma_inst pinstanc
```

où `p` est le paramètre de position (6.4.2), `lemma` est le lemme dont on a identifié la dépendance (7.3), `plemma` est la preuve de ce lemme, `lemma_inst` est l'instance et `pinstanc` est un élément du type `lemma -> interp lemma_inst`, c'est-à-dire une preuve que le lemme implique l'instance. Le problème (2) de la partie précédente est donc résolu: le lemme est directement transmis à la règle `Forallinst`.

Une étape de la fonction `checker` correspondant à une telle *crule* enregistre la clause `lemma_inst` à l'emplacement `p` de l'état. On peut facilement prouver que `lemma_inst` est valide, et même indépendamment des clauses contenues par l'état. En effet, il suffit d'appliquer `pinstanc` à `plemma`. On a donc rétabli la preuve du théorème de correction.

Le problème est en fait déplacé puisqu'il faut maintenant, pour construire une *crule*, donner un élément du type `lemma -> interp lemma_inst` qu'on appellera preuve d'instanciation. Cette preuve est donnée à l'aide d'une coupure (tactique `assert` en Coq), ce qui nous permet de donner les preuves d'instanciation dans un deuxième temps.

8.2 Preuves d'instanciation

La structure d'une instance peut différer de celle du lemme pour plusieurs raisons. Pour chacune de ces raisons nous verrons comment obtenir tout de même une preuve d'instanciation. Grâce à ces résultats on peut définir une tactique qui permet de trouver automatiquement n'importe quelle preuve d'instanciation (voir annexe C).

8.2.1 Preuve automatique d'une instanciation

Le lemme Coq suivant correspond directement à l'instanciation du lemme donné:

```
Lemma instance_c P (c : Z):
  (forall x, P x) ->
  P c.
```

De tels buts peuvent être prouvés automatiquement par la tactique `auto`. Cependant, ce n'est plus le cas lorsque le but est légèrement modifié. Par exemple, le lemme suivant ne peut pas être prouvé directement par la tactique `auto`:

```
Lemma instance_3 f (c : Z):
  (forall x, f x =? f c) ->
  f c =? f 3.
```

Il faut donc remettre le but dans une forme qui correspond à celle du lemme. Dans le cas ci-dessus il faut utiliser la symétrie de l'égalité avant d'appliquer la tactique `auto`.

8.2.2 Différence liée à la symétrie de l'égalité

Les lemmes qui apparaissent dans les certificats de veriT peuvent avoir subi des modifications (7.3.2), ces modifications se retrouvent dans les preuves d'instanciation des lemmes. Par exemple, à partir du lemme Coq `forall x, f x =? f c` où `c` est une constante entière, le certificat de veriT peut contenir la règle:

$$id1 : (input\ (\forall\ x,\ f\ c = f\ x))$$

Une règle d'instanciation correspondant à ce lemme sera de la forme:

$$id2 : (forall\_inst\ (\neg(\forall\ x,\ f\ c = f\ x)\ (f\ c = f\ 3)))$$

Il s'agit donc de trouver une preuve de `instance_3`. Ce problème n'a été résolu que partiellement en remarquant que la plupart du temps, si une des égalités est inversée, alors toutes les égalités sont inversées. On écrit donc une tactique qui inverse toutes les égalités et pour chaque cas on essaye de résoudre le but en utilisant cette tactique et en ne l'utilisant pas. Le cas général pourrait être traité de manière naïve en temps exponentiel. Nous laissons une implémentation efficace pour les travaux futurs.

8.2.3 Différence *impl_split*

Lorsqu'un lemme est de la forme:

$$id1 : (input\ (\forall\ x,\ f\ x \Rightarrow b))$$

la *trule* $forall\_inst$ peut être:

$$id2 : (forall\_inst\ (\neg(\forall\ x,\ f\ x \Rightarrow b)\ (\neg(f\ c) \vee b)))$$

au lieu de la *trule* attendue:

$$id2 : (forall\_inst\ (\neg(\forall\ x,\ f\ x \Rightarrow b)\ (f\ c \Rightarrow b)))$$

Cette différence se retrouve directement dans la preuve d'instanciation associée et peut être résolue en ajoutant un lemme à la base de lemmes `Resolve`. La tactique `auto`, qui utilise cette base de lemmes, trouvera automatiquement la preuve pour ce type de différence entre lemme et instance.

```
Lemma impl_split a b:
  implb a b -> orb (negb a) b.
Proof. intro. destruct a; destruct b; trivial. Qed.

Hint Resolve impl_split.
```

9 Utilisation de la tactique `verit` avec des lemmes

On définit une nouvelle tactique : `verit_base`. Cette tactique peut prendre en argument des preuves de lemmes à rajouter. Par exemple

```
Lemma instance_double f k:
  (forall x, f (x + 1) =? f x + k) ->
  forall x, f (x + 2) =? f x + 2 * k.
Proof. intro f_k_linear. verit_base f_k_linear.
```

Cette tactique laisse à l'utilisateur les 2 preuves d'instanciation. On peut terminer la preuve grâce à la tactique `vauto` définie d'après les résultats de la partie précédente :

```
Proof. intro f_k_linear. verit_base f_k_linear. vauto. vauto. Qed
```

Plus généralement, lorsqu'on veut rajouter les lemmes prouvés par `H1`, ..., `Hn`, on utilisera :

`verit_base H1 ... Hn; vauto.`

On définit la tactique `verit` comme `verit_base; vauto` ce qui nous permet de conserver le comportement de cette tactique par rapport à la version antérieure de SMTCoq lorsque l'on n'utilise pas de lemmes supplémentaires. On définit également une liste qui sera toujours ajoutée à `verit_base`. Initialement cette liste est vide mais la commande `Add_lemmas H1 ... Hn` la concatène avec `H1`, ...,`Hn`. Pour finir la commande `Clear_lemmas` réinitialise la liste.
Le dépôt Git `github.com/QGarchery/smtcoq-1` contient d'autres exemples donnés dans les fichiers `examples/Example.v` et `unit-tests/Tests_verit.v`.

10 Travaux connexes et conclusion

Ce stage s'inscrit dans une approche automatique de la démonstration dans un assistant de preuve et porte plus précisément sur SMTCoq.
D'autres outils sont disponibles dans ce cadre, à commencer par Coqhammer [6] qui est aussi un *plugin* pour Coq qui utilise des prouveurs automatiques. Une première différence est que SMTCoq propose aussi une commande de vérification, ce qui permet d'améliorer la confiance accordée aux prouveurs automatiques (5.2.2). Une autre différence est que lors de la reconstruction de la preuve, Coqhammer exrait la liste les lemmes qui apparaissent dans le certificat et n'utilise rien d'autre que cette liste du certificat. Ainsi, il y a une recherche qui est faite par des tactiques en Coq et qui permet de retrouver la preuve. Cette méthode est plus robuste que celle de SMTCoq vis-à-vis des prouveurs automatiques mais demande de chercher à nouveau la preuve et est donc plus coûteuse. En revanche, nous avons effectué les preuves d'instanciation à l'aide de tactiques Coq (8.2), ce qui se rapproche des techniques utilisées par Coqhammer.
Toujours dans l'approche sceptique, Sledgehammer [5] est une interface avec des prouveurs automatiques qui est développée pour l'assistant de preuve Isabelle. L'idée d'un sélectionneur de lemmes qui apprend quels lemmes il est judicieux d'envoyer aux prouveurs automatiques [4] vient de Sledgehammer. On peut aussi mentionner des approches autarciques de la démonstration automatique comme la tactique `unsat` [11] qui utilise la réflexion calculatoire ou encore la tactique `metis` en Isabelle.

En résumé, ce stage a permis d'améliorer l'expressivité de SMTCoq tout en restant dans un cadre qui assure la correction de la méthode. Cette amélioration de l'expressivité a été confirmée par de nouveaux tests. Enfin, ce stage ouvre de nouvelles pistes de travail (1.5) comme l'amélioration de l'efficacité, la mise en place d'un sélectionneur de lemmes, l'extension des cas d'application ou encore l'utilisation de SMTCoq pour la certification Why3.

Remerciements

À la suite de mon stage au LRI d'Orsay, j'adresse tous mes remerciements

À mes professeurs du Master Parisien de Recherche en Informatique qui m'ont donné la possibilité de trouver un stage correspondant tout à fait à mes attentes,

À mes maîtres de stage, madame Chantal Keller, maître de Conférences de l'Université Paris-Sud et monsieur Valentin Blot, post-doctorant de l'Université Paris-Sud, pour leur disponibilité et leur enthousiasme,
À madame Chantal Keller qui a bien voulu m'initier au sujet de sa recherche et m'a donné de précieux conseils à différents niveaux,
À monsieur Valentin Blot de son aide fréquente, si utile pour la bonne compréhension du sujet, et des discussions passionnantes que nous avons eues qui m'ont ouvert d'autres perspectives,

À toute l'équipe VALS qui m'a accueilli et en particulier à monsieur Claude Marché, madame Chantal Keller et monsieur Andrei Paskevich, que je serai très heureux de retrouver en tant qu'encadrants de ma thèse.

Bibliographie

[1] Michael Armand, Germain Faure, Benjamin Grégoire, Chantal Keller, Laurent Théry, and Benjamin Werner. A modular integration of sat/smt solvers to coq through proof witnesses. In Jean-Pierre Jouannaud and Zhong Shao, editors, *Certified Programs and Proofs*, pages 135–150, Berlin, Heidelberg, 2011. Springer Berlin Heidelberg.

[2] Michaël Armand, Benjamin Grégoire, Arnaud Spiwack, and Laurent Théry. Extending coq with imperative features and its application to SAT verification. In *Interactive Theorem Proving, First International Conference, ITP 2010, Edinburgh, UK, July 11-14, 2010. Proceedings*, pages 83–98, 2010.

[3] Clark Barrett, Pascal Fontaine, and Cesare Tinelli. The SMT-LIB Standard: Version 2.6. Technical report, Department of Computer Science, The University of Iowa, 2017. Available at `www.SMT-LIB.org`.

[4] Jasmin Christian Blanchette, David Greenaway, Cezary Kaliszyk, Daniel Kühlwein, and Josef Urban. A learning-based fact selector for isabelle/hol. *J. Autom. Reasoning*, 57(3):219–244, 2016.

[5] Jasmin Christian Blanchette and Lawrence C. Paulson. *Hammering Away, A User's Guide to Sledgehammer for Isabelle/HOL*, 2017.

[6] Lukasz Czajka and Cezary Kaliszyk. Hammer for coq: Automation for dependent type theory. *J. Autom. Reasoning*, 61(1-4):423–453, 2018.

[7] Jean-Christophe Filliâtre and Andrei Paskevich. Why3 - where programs meet provers. In *Programming Languages and Systems - 22nd European Symposium on Programming, ESOP 2013, Held as Part of the European Joint Conferences on Theory and Practice of Software, ETAPS 2013, Rome, Italy, March 16-24, 2013. Proceedings*, pages 125–128, 2013.

[8] Georges Gonthier, Assia Mahboubi, and Enrico Tassi. A Small Scale Reflection Extension for the Coq system. Research Report RR-6455, Inria Saclay Ile de France, 2016.

[9] Chantal Keller. *A Matter of Trust: Skeptical Communication Between Coq and External Provers. (Question de confiance : communication sceptique entre Coq et des prouveurs externes)*. PhD thesis, École Polytechnique, Palaiseau, France, 2013.

[10] Xavier Leroy. Formal verification of a realistic compiler. *Commun. ACM*, 52(7):107–115, 2009.

[11] Stéphane Lescuyer. *Formalizing and Implementing a Reflexive Tactic for Automated Deduction in Coq. (Formalisation et developpement d'une tactique reflexive pour la demonstration automatique en coq)*. PhD thesis, University of Paris-Sud, Orsay, France, 2011.

[12] Robert Nieuwenhuis, Albert Oliveras, and Cesare Tinelli. Solving SAT and SAT modulo theories: From an abstract davis–putnam–logemann–loveland procedure to dpll(T). *J. ACM*, 53(6):937–977, 2006.

[13] Christine Paulin-Mohring. Introduction to the Calculus of Inductive Constructions. In Bruno Woltzenlogel Paleo and David Delahaye, editors, *All about Proofs, Proofs for All*, volume 55 of *Studies in Logic (Mathematical logic and foundations)*. College Publications, January 2015.

Annexe A: Formules booléennes disjonctives

Le fichier Coq `ortree.v` à la racine du répertoire SMTCoq contient:

```
Require Import Bool.

Inductive OrTree :=
  Bool (b : bool)
| Or (left: OrTree) (right: OrTree).

Inductive Interp : OrTree -> Prop :=
  InterpBool :
    Interp (Bool true)
| InterpOrLeft t1 t2 :
    Interp t1 -> Interp (Or t1 t2)
| InterpOrRight t1 t2 :
    Interp t2 -> Interp (Or t1 t2).

Definition t :=
  Or (Or (Bool false) (Bool true)) (Or (Bool false) (Bool false)).

Lemma Interp_t : Interp t.
Proof.
  apply InterpOrLeft.
  apply InterpOrRight.
  apply InterpBool.
Qed.

Fixpoint interp (t : OrTree) :=
  match t with
    Bool b => b
  | Or t1 t2 => interp t1 || interp t2
  end.

Proposition Interp_equiv_interp t :
  Interp t <-> interp t = true.

Proof.
  induction t as [a | t1 IHt1 t2 IHt2]; simpl.
  -split; intro H.
   +now inversion H.
   +rewrite H. apply InterpBool.
  -split; intro H.
   +inversion H.
    now apply IHt1 in H1; rewrite H1.
    apply IHt2 in H1; rewrite H1. apply orb_true_r.
   +apply orb_true_iff in H. destruct H.
    now apply InterpOrLeft, IHt1.
    now apply InterpOrRight, IHt2.
Qed.

Lemma interp_t : interp t = true.
Proof.
  reflexivity.
Qed.
```

```
Fixpoint append t1 t2 :=
  match t1 with
  | Bool _ => Or t1 t2
  | Or t11 t12 => Or t11 (append t12 t2)
  end.

Fixpoint peigne (t : OrTree) :=
  match t with
  | Bool n => t
  | Or t1 t2 => let pt1 := peigne t1 in
                let pt2 := peigne t2 in
                append pt1 pt2
  end.

Inductive eqt : OrTree -> OrTree -> Prop :=
  refl t: eqt t t
| sym t1 t2: eqt t1 t2 -> eqt t2 t1
| assoc t1 t2 t3: eqt (Or t1 (Or t2 t3)) (Or (Or t1 t2) t3)
| congru ta1 ta2 tb1 tb2: eqt ta1 tb1 -> eqt ta2 tb2 ->
                          eqt (Or ta1 ta2) (Or tb1 tb2)
| trans t1 t2 t3: eqt t1 t2 -> eqt t2 t3 -> eqt t1 t3.

Lemma eqt_correct t1 t2:
  eqt t1 t2 -> interp t1 = interp t2.
Proof.
  intro eq12. induction eq12; simpl.
  -reflexivity.
  -auto.
  -apply orb_assoc.
  -now rewrite IHeq12_1, IHeq12_2.
  -now rewrite IHeq12_1.
Qed.

Lemma append_eqt t1 t2:
  eqt (append t1 t2) (Or t1 t2).
Proof.
  revert t2. induction t1; intro t2; simpl.
  -apply refl.
  -eapply trans. eapply congru. apply refl. apply IHt1_2.
   apply assoc.
Qed.

Lemma peigne_eqt t:
  eqt (peigne t) t.
Proof.
  induction t; simpl.
  -apply refl.
  -eapply trans. apply append_eqt. now apply congru.
Qed.

Lemma peigne_correct t:
  interp (peigne t) = interp t.
Proof.
  apply eqt_correct. now apply peigne_eqt.
Qed.
```

```
Ltac reify A := match A with
              | orb ?X ?Y => let rx := reify X in
                             let ry := reify Y in
                             constr:(Or rx ry)
              | ?X => constr:(Bool X) end.

Ltac peignify :=
  match goal with
  | [ |- ?A = ?B] =>
    let a := reify A in
    let b := reify B in
    change A with (interp a);
    change B with (interp b);
    rewrite <- (peigne_correct a);
    rewrite <- (peigne_correct b);
    simpl
  end.

Lemma or_tree_equality b1 b2 b3 b4:
  (b1 || b2) || (b3 || b4) = b1 || ((b2 || b3) || b4).
Proof.
  peignify. reflexivity.
Qed.
```

Annexe B: Fichier SMT-LIB et certificat veriT d'une instanciation

Dans le répertoire `unit-tests` de SMTCoq, la commande

```
./runverit.sh instance2.smt2
```

appelle veriT sur le fichier SMT-LIB `instance2.smt2` suivant:

```
(set-logic UFLIA)
(declare-fun f (Int) Int)
(assert (not (= (f 3) (+ (f 2) 7))))
(assert (forall ( (x Int) ) (= (f (+ x 1)) (+ (f x) 7))))
(check-sat)
(exit)
```

On obtient *unsat* et le fichier de certificat `instance2.vtlog` contenant:

```
1:(input ((not #1:(= #2:(f 3) #3:(+ #4:(f 2) 7)))))
2:(input (#5:(forall ( (x Int) ) #6:(= #7:(f #8:(+ x 1)) #9:(+ #10:(f x) 7)))))
3:(tmp_betared (#11:(forall ( (@vr0 Int) ) #12:(= #13:(f #14:(+ @vr0 1)) #15:(+ #16:(f
    @vr0) 7)))) 2)
4:(tmp_qnt_tidy (#17:(forall ( (@vr1 Int) ) #18:(= #19:(f #20:(+ @vr1 1)) #21:(+ #22:(f
    @vr1) 7)))) 3)
5:(forall_inst (#23:(or (not #17) #24:(= #3 #25:(f #26:(+ 2 1))))))
6:(or ((not #17) #24) 5)
7:(resolution (#24) 6 4)
8:(eq_transitive ((not #27:(= #2 #25)) (not #24) #1))
9:(eq_congruent ((not #28:(= 3 #26)) #27))
10:(resolution ((not #24) #1 (not #28)) 8 9)
11:(resolution ((not #28)) 10 1 7)
12:(la_disequality (#29:(or #28 (not #30:(<= 3 #26)) (not #31:(<= #26 3)))))
```

```
13:(or (#28 (not #30) (not #31)) 12)
14:(resolution ((not #30) (not #31)) 13 11)
15:(la_generic (#31))
16:(resolution ((not #30)) 14 15)
17:(la_generic (#30))
18:(resolution () 17 16)
```

Annexe C: Automatisation des preuves d'instanciation

Dans le fichier Coq `src/SMTCoq.v`, on écrit:

```
(* verit silently transforms an <implb a b> into a <or (not a) b> when
 instantiating a quantified theorem with <implb> *)
Lemma impl_split a b:
  implb a b = true -> orb (negb a) b = true.
Proof.
  intro H.
  destruct a; destruct b; trivial.
(* alternatively we could do <now verit_base H.> but it forces us to have veriT
   installed when we compile SMTCoq. *)
Qed.

Hint Resolve impl_split.

(* verit silently transforms an <implb (a || b) c> into a <or (not a) c>
   or into a <or (not b) c> when instantiating such a quantified theorem *)
Lemma impl_or_split_right a b c:
  implb (a || b) c -> negb b || c.
Proof.
  intro H.
  destruct a; destruct c; intuition.
Qed.

Lemma impl_or_split_left a b c:
  implb (a || b) c -> negb a || c.
Proof.
  intro H.
  destruct a; destruct c; intuition.
Qed.

(* verit considers equality modulo its symmetry, so we have to recover the
   right direction in the instances of the theorems *)
Definition hidden_eq a b := a =? b.
Ltac all_rew :=
  repeat match goal with
         | [ |- context [ ?A =? ?B]] =>
           change (A =? B) with (hidden_eq A B)
         end;
  repeat match goal with
         | [ |- context [ hidden_eq ?A ?B] ] =>
           replace (hidden_eq A B) with (B =? A);
           [ | now rewrite Z.eqb_sym]
         end.
```

```
(* An automatic tactic that takes into account all those transformations *)
Ltac vauto :=
  try (let H := fresh "H" in
       intro H; try (all_rew; apply H);
       match goal with
       | [ |- is_true (negb ?A || ?B) ] =>
         try (eapply impl_or_split_right; apply H);
         eapply impl_or_split_left; apply H
       end;
       apply H);
  auto.

Ltac verit :=
  verit_base; vauto.
```